

# NAG Library Routine Document

## G01KQF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G01KQF returns a number of values of the probability density function (PDF), or its logarithm, for the Normal (Gaussian) distributions.

### 2 Specification

SUBROUTINE G01KQF (ILOG, LX, X, LXMU, XMU, LXSTD, XSTD, PDF, IVALID, &  
IFAIL)

INTEGER ILOG, LX, LXMU, LXSTD, IVALID(\*), IFAIL  
REAL (KIND=nag\_wp) X(LX), XMU(LXMU), XSTD(LXSTD), PDF(\*)

### 3 Description

The Normal distribution with mean  $\mu_i$ , variance  $\sigma_i^2$ ; has probability density function (PDF)

$$f(x_i, \mu_i, \sigma_i) = \frac{1}{\sigma_i \sqrt{2\pi}} e^{-(x_i - \mu_i)^2 / 2\sigma_i^2}, \quad \sigma_i > 0.$$

The input arrays to this routine are designed to allow maximum flexibility in the supply of vector arguments by re-using elements of any arrays that are shorter than the total number of evaluations required. See Section 2.6 in the G01 Chapter Introduction for further information.

### 4 References

None.

### 5 Arguments

1: ILOG – INTEGER *Input*

*On entry:* the value of ILOG determines whether the logarithmic value is returned in PDF.

ILOG = 0

$f(x_i, \mu_i, \sigma_i)$ , the probability density function is returned.

ILOG = 1

$\log(f(x_i, \mu_i, \sigma_i))$ , the logarithm of the probability density function is returned.

*Constraint:* ILOG = 0 or 1.

2: LX – INTEGER *Input*

*On entry:* the length of the array X.

*Constraint:* LX > 0.

3: X(LX) – REAL (KIND=nag\_wp) array *Input*

*On entry:*  $x_i$ , the values at which the PDF is to be evaluated with  $x_i = X(j)$ ,  $j = ((i - 1) \bmod LX) + 1$ , for  $i = 1, 2, \dots, \max(LX, LXSTD, LXMU)$ .

- 4: LXMU – INTEGER *Input*  
*On entry:* the length of the array XMU.  
*Constraint:* LXMU > 0.
- 5: XMU(LXMU) – REAL (KIND=nag\_wp) array *Input*  
*On entry:*  $\mu_i$ , the means with  $\mu_i = \text{XMU}(j)$ ,  $j = ((i - 1) \bmod \text{LXMU}) + 1$ .
- 6: LXSTD – INTEGER *Input*  
*On entry:* the length of the array XSTD.  
*Constraint:* LXSTD > 0.
- 7: XSTD(LXSTD) – REAL (KIND=nag\_wp) array *Input*  
*On entry:*  $\sigma_i$ , the standard deviations with  $\sigma_i = \text{XSTD}(j)$ ,  $j = ((i - 1) \bmod \text{LXSTD}) + 1$ .  
*Constraint:*  $\text{XSTD}(j) \geq 0.0$ , for  $j = 1, 2, \dots, \text{LXSTD}$ .
- 8: PDF(\*) – REAL (KIND=nag\_wp) array *Output*  
**Note:** the dimension of the array PDF must be at least  $\max(\text{LX}, \text{LXSTD}, \text{LXMU})$ .  
*On exit:*  $f(x_i, \mu_i, \sigma_i)$  or  $\log(f(x_i, \mu_i, \sigma_i))$ .
- 9: IVALID(\*) – INTEGER array *Output*  
**Note:** the dimension of the array IVALID must be at least  $\max(\text{LX}, \text{LXSTD}, \text{LXMU})$ .  
*On exit:* IVALID( $i$ ) indicates any errors with the input arguments, with  
 IVALID( $i$ ) = 0  
     No error.  
 IVALID( $i$ ) = 1  
      $\sigma_i < 0$ .
- 10: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, at least one value of XSTD was invalid.  
 Check IVALID for more information.

IFAIL = 2

On entry, ILOG =  $\langle value \rangle$ .  
Constraint: ILOG = 0 or 1.

IFAIL = 3

On entry, array size =  $\langle value \rangle$ .  
Constraint: LX > 0.

IFAIL = 4

On entry, array size =  $\langle value \rangle$ .  
Constraint: LXMU > 0.

IFAIL = 5

On entry, array size =  $\langle value \rangle$ .  
Constraint: LXSTD > 0.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.  
See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.  
See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.  
See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

Not applicable.

## 8 Parallelism and Performance

G01KQF is not threaded in any implementation.

## 9 Further Comments

None.

## 10 Example

This example prints the value of the Normal distribution PDF at four different points  $x_i$  with differing  $\mu_i$  and  $\sigma_i$ .

### 10.1 Program Text

```

Program g01kqfe
!   G01KQF Example Program Text

!   Mark 26 Release. NAG Copyright 2016.

!   .. Use Statements ..
   Use nag_library, Only: g01kqf, nag_wp

```

```

!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, ilog, lout, lx, lxmu,      &
                                   lxstd
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: pdf(:), x(:), xmu(:), xstd(:)
      Integer, Allocatable          :: ivalid(:)
!      .. Intrinsic Procedures ..
      Intrinsic                    :: max, mod, repeat
!      .. Executable Statements ..
      Write (nout,*) 'G01KQF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Read in flag indicating whether logs are required
      Read (nin,*) ilog

!      Read in the input vectors
      Read (nin,*) lx
      Allocate (x(lx))
      Read (nin,*) x(1:lx)

      Read (nin,*) lxmu
      Allocate (xmu(lxmu))
      Read (nin,*) xmu(1:lxmu)

      Read (nin,*) lxstd
      Allocate (xstd(lxstd))
      Read (nin,*) xstd(1:lxstd)

!      Allocate memory for output
      lout = max(lx,lxmu,lxstd)
      Allocate (pdf(lout),ivalid(lout))

!      Calculate the PDF
      ifail = -1
      Call g01kqf(ilog,lx,x,lxmu,xmu,lxstd,xstd,pdf,ivalid,ifail)

      If (ifail==0 .Or. ifail==1) Then
!      Display titles
      Write (nout,*) '      X      XMU      XSTD      PDF      IVALID'
      Write (nout,*) repeat('-',50)

!      Display results
      Do i = 1, lout
         Write (nout,99999) x(mod(i-1,lx)+1), xmu(mod(i-1,lxmu)+1),      &
                               xstd(mod(i-1,lxstd)+1), pdf(i), ivalid(i)
      End Do
      End If

99999 Format (1X,3(F6.2,4X),E10.3,4X,I3)
      End Program g01kqfe

```

## 10.2 Program Data

```

G01KQF Example Program Data
0          :: ILOG
4          :: LX
1.0 4.0 0.1 1.0      :: X
4          :: LXMU
0.0 2.0 0.0 0.0     :: XMU
4          :: LXSTD
1.0 1.0 0.01 10.0   :: XSTD

```

### 10.3 Program Results

G01KQF Example Program Results

X	XMU	XSTD	PDF	IVALID
1.00	0.00	1.00	0.242E+00	0
4.00	2.00	1.00	0.540E-01	0
0.10	0.00	0.01	0.769E-20	0
1.00	0.00	10.00	0.397E-01	0

